

# Maple: Производни и критични точки

## Въведение

Ние знаем, че Maple може лесно да изпълнява символни алгебрични операции. Поради този факт Maple е идеален пакет за извършване на символни пресмятания, свързани с математическия анализ. В тази глава ние ще демонстрираме как програмата може да помогне за намиране на производни на функция, определяне на критичните точки на функцията (ако те съществуват) и как може да се класифицират критичните точки.

## Изчисляване на производна на функция

Дефинираме функцията  $f$  с командата

```
> f := x^3*(x^2-1);
```

$$f := x^3(x^2 - 1)$$

Производната на функцията се изчислява чрез командата `diff`. За да научим повече за това как коректно да използваме `diff` може да видим съответния помощен файл на Maple. За Целта пишем въпросителен знак, следван непосредствено (без интервал) от името на командата и натискаме **Enter**. Например:

```
> ?diff
```

Помощният файл се появява в нов прозорец, показващ правилния синтаксис както и няколко примера как да се използва командата.

За да изчислим първата производна на функцията, използвайки командата `diff`, трябва да укажем коя функция ще диференцираме и спрямо коя променлива ще диференцираме. Ако искаме да диференцираме функцията  $f$ , дефинирана по-горе по отношение на  $x$  трябва да напишем

```
> diff(f,x);
```

$$3x^2(x^2 - 1) + 2x^4$$

Има два начина по които можем да изчислим втората производна на функцията  $f$ . Първият начин е да присвоим име, например  $g$ , на резултата от първата производна и тогава да изчислим  $\text{diff}(g,x)$  както е показано по-долу:

```
> g:=diff(f,x);
```

$$g := 3x^2(x^2 - 1) + 2x^4$$

```
> diff(g,x);
```

$$6x(x^2 - 1) + 14x^3$$

Вторият начин е по-директен – в този случай просто инструктираме Maple да диференцира функцията  $f$  два пъти и получаваме втората производна.

```
> diff(f,x,x);
```

$$6x(x^2 - 1) + 14x^3$$

Можем да изчислим която и да е производна на функцията. Командата, която дава четвъртата производна е следната:

```
> diff(f,x$4);
```

$$120x$$

Ако сравним резултата, който бихме получили с командата  $\text{diff}(f,x,x,x,x)$ , ще видим че той ще е точно същия. Използването на  $x\$4$  позволява по-бързо да запишем исканото, но е също така синтактично по-правилно, тъй като е по-лесно да се промени до по-голям реда на производната, отколкото по другия начин.

В случая на функция на няколко променливи, Maple също може да изчислява производните достатъчно лесно. Същественото, когато въвеждаме функция на няколко променливи е определянето на променливата, по която ще се диференцира

```
> h := 3*x*y^2*(x^2-4*y);
```

$$h := 3xy^2(x^2 - 4y)$$

```
> diff(h,x);
```

$$3y^2(x^2 - 4y) + 6x^2y^2$$

```
> diff(h,y);
```

$$6xy(x^2 - 4y) - 12xy^2$$

## Изчисляване на критични точки


**Критична точка** е точка от дефиниционната област на функцията, в която тя нейната производна е нула. Иначе казано, наклонът на функцията, изчислен в тази точка, е равен на нула. Преди да изчислим критичните точки на функцията е полезно да видим първо нейната графика, за да можем да изчислим приблизително стойността на дадена критична точка. Ще използваме функцията **f**, дефинирана в предната секция

```
> f := x^3*(x^2-1);
```

$$f := x^3(x^2 - 1)$$

За да нарисуваме графиката на функцията използваме командата **plot**. Тази команда има два аргумента: първият е функцията, която чертаем, а вторият е променливата със съответния интервал. Например, за да нарисуваме функцията **f** в интервала  $[-1,+1]$  записваме:

```
> plot(f,x=-1..1);
```



Фигура 1: Графика на  $y = x^3(x^2 - 1)$  в интервала  $[-1,+1]$ .

Разглеждането на графиката подсказва, че има три точки, в които допирателната към графиката на функцията може да е успоредна на оста  $Ox$ :

1. Точка от интервала  $(-1, 0)$ , която е близка е  $x = -0.75$ .

2. Близка точка  $x = 0$ .

3. Точка в интервала  $(0, 1)$ , която е близка е  $x = 0.75$ .

За да изчислим критичните точки диференцираме функцията и търсим къде е равна на нула. Ще присвоим име `df` на производната на функцията `f`.

```
> df := diff(f,x);
```

$$df := 3x^2(x^2 - 1) + 2x^4$$

Решавайки уравнението `df=0` с командата `solve`, получаваме критичните точки. Maple, по подразбиране, дава критичните точки като дроби. Ако искаме да видим отговорите като десетични числа, трябва да използваме командата `evalf`.

```
> cp := evalf(solve(df=0,x));
```

$$cp := 0., 0., 0.7745966692, -0.7745966692$$

Maple ни дава четири критични точки, но при разглеждането е ясно, че две от тези критични точки, намиращи се в началото на координатната система, съвпадат. Можем да изчислим критичната стойност на всяка от критичните точки, чрез заместване на всяка от точка във формулата на `f`. Да припомним, че `cp` се отнася до всичките четири отговора, дадени от Maple и за да получим стойността във всяка от отделните критични точки ще използваме квадратни скоби, т.е. `cp[1]` ще върне стойността на първата критична точка, от списъка на `cp`, т.е. 0.

```
> evalf( subs(x=cp[1], f) );
```

$$-0.$$

Следователно, координатите на първата критична точка са  $(0, 0)$ . `cp[2]` ще даде същия резултат и няма нужда да го разглеждаме. Критичните стойности, съответстващи на другите две точки `cp[3]` и `cp[4]` се изчисляват по същия начин.

```
> evalf( subs(x=cp[3],f) );
```

$$-0.1859032006$$

```
> evalf( subs(x=cp[4],f) );
```

$$0.1859032006$$

Следователно, координатите на последните две критични точки са

(0.7745966692, -0.1859032006) и (-0.7745966692, 0.1859032006). За да проверим дали критичните точки са локален максимум, минимум или инфлексни точки, трябва да ги заместим във вторите производни. Тестът на втората производна се състои в следното:

Ако функцията  $f$  е два пъти диференцируема в околност на фиксирана точка  $x$  и  $f'(x) = 0$ , тогава

- Ако  $f''(x) < 0$ , то  $f$  има локален максимум в точката  $x$ .
- Ако  $f''(x) > 0$ , то  $f$  има локален минимум в точката  $x$
- Ако  $f''(x) = 0$ , то тестът с втората производна е неопределен за точката  $x$  – тя може да е инфлексна точка, но може и да не е.

Ако  $f''(x) = 0$ , тогава един лесен начин да проверим дали критичната точка е инфлексна точка е да пресметнем  $f''$  в две точки в близост до критичната точка, по една от всяка страна на критичната точка, и да видим дали те имат различни знаци. Ако да, тогава критичната точка, която анализираме е инфлексна точка.

За нашия пример, нека изчислим втората производна на функцията  $f$  и да пресметнем стойностите ѝ във всяка от критичните точки

```
> d2f := diff(df,x);  
d2f := 6 x (x2 - 1) + 14 x3
```

Пресмятаме втората производна в критичната точка `cp[1]`, т.е.  $x = 0$ :

```
> evalf(subs(x=cp[1],d2f));  
0.
```

Този тест е неопределен и затова трябва да пресметнем  $f''$  в две точки близо до  $x = 0$ . Двете точки ще изберем например така:  $x = -0.15$  и  $x = 0.15$ .

```
> evalf(subs(x=0.15,d2f));  
-0.832500  
  
> evalf(subs(x=-0.15,d2f));  
0.832500
```

От това, че знаците в тези две точки са с различни знаци следва, че точката  $x = 0$  е инфлексна точка.

По-нататък пресмятаме втората производна в критичната точка  $cp[3]$ , т.е.  $x = 0.7745966692$ .

```
> evalf(subs(x=cp[3],d2f));  
4.647580014
```

Втората производна е положителна, което означава че критичната точка  $x = 0.7745966692$  е локален минимум.

Накрая пресмятаме втората производна в критичната точка  $cp[4]$ , т.е.  $x = -0.7745966692$ :

```
> evalf(subs(x=cp[4],d2f));  
-4.647580014
```

Втората производна е отрицателна, което означава, че критичната точка  $x = -0.7745966692$  е локален максимум. Така изчислихме и класифицирахме критичните точки на дадената функция.